東京医大2017年　入試問題

1 平面上に六点 O，A_1，A_2，B_1，B_2，P があり，点 O を始点とする有向線分が表すベクトルの内積の値が

$$\overrightarrow{OA_1}\cdot\overrightarrow{OB_1}=3,\quad \overrightarrow{OA_1}\cdot\overrightarrow{OB_2}=4,\quad \overrightarrow{OA_1}\cdot\overrightarrow{OP}=14,$$
$$\overrightarrow{OA_2}\cdot\overrightarrow{OB_1}=5,\quad \overrightarrow{OA_2}\cdot\overrightarrow{OB_2}=7,\quad \overrightarrow{OA_2}\cdot\overrightarrow{OP}=23,$$
$$\overrightarrow{OP}\cdot\overrightarrow{OB_1}=22,\quad \overrightarrow{OP}\cdot\overrightarrow{OB_2}=35$$

であった．このとき，$\overrightarrow{A_1A_2}\cdot\overrightarrow{PB_1}=\boxed{アイ}$ であり，線分 OP の長さは $OP=\sqrt{\boxed{ウエ}}$ である．

```
ClearAll["Global`*"]
A1 = {a, 0}; A2 = {a3, a4}; B1 = {b1, b2}; B2 = {b3, b4}; p = {x, y};
eqs = A1.B1 == 3 && A1.B2 == 4 && A2.B1 == 5 &&
   A2.B2 == 7 && A1.p == 14 && A2.p == 23 && B1.p == 22 && B2.p == 35;

Exists[{a, a3, a4, b1, b2, b3, b4, x, y}, eqs && (A2 - A1).(B1 - p) == k1];
Resolve[%, Reals]
Exists[{a, a3, a4, b1, b2, b3, b4, x, y}, eqs && x^2 + y^2 == k2];
Resolve[%, Reals]
```

k1 == -7

k2 == 97

---

2 (1) a，b，c を定数とする．関数 $f(x)=x^5+ax^4+bx^3+cx^2$ が，$x=0$ において極大値 0 をとり，$x=5$ において極小値 0 をとるときについて考える．
このとき，定数 b の値の範囲は $\boxed{アイ}<b<\boxed{ウエ}$ である．
また，このときの $f(x)$ は $x=0$ 以外でも極大値をとり，その極大値 M の範囲は $\boxed{オ}<M<\boxed{カキク}$ である．

(2) 関数 $f(x)=(ax)^{-\frac{(bx)^{cx}}{3}}$ の $x=1$ における微分係数は $a=8$，$b=e^{-1}$，$c=-\log 2$ であるとき $f'(1)=\dfrac{\boxed{ケコ}}{\boxed{サ}}$ である．ただし，対数は e を底とする自然対数とする．

```
(*1*)
f[x_] := x^5 + a x^4 + b x^3 + c x^2;
cons = f'[0] == 0 ∧ f[0] == 0 ∧ f''[0] < 0 ∧ f'[5] == 0 ∧ f[5] == 0 ∧ f''[5] > 0;
(*conditions*)
Exists[{a, c}, cons];
Resolve[%, Reals] // Simplify
Exists[{a, b, c, p}, cons && f'[p] == 0 && f[p] == M && f''[p] < 0 && p ≠ 0];
Resolve[%, Reals] // Simplify
```

25 < b < 75

0 < M < 108

```
(*2*) ClearAll["Global`*"]
f[x_] = ((a x) ^ (- (b x) ^ (c x) / 3)) /. {a → 8, b → 1 / E, c → - Log[2]};
f'[1]
```

$-\frac{1}{6}$

---

3 条件

$$a_1 = 1, \quad a_{n+1} = a_n \cos \frac{n^2 \pi}{3}$$

によって定められる数列 $\{a_n\}$ について考える.

(1) $a_6 = \dfrac{\boxed{アイ}}{\boxed{ウエ}}$ であり, $\displaystyle\sum_{n=1}^{6} a_n = \dfrac{\boxed{オカ}}{\boxed{キク}}$ である.

(2) $\log_2 |a_{70}| = \boxed{ケコサ}$ であり, $\displaystyle\sum_{k=1}^{\infty} a_{6k+1} = \dfrac{\boxed{シス}}{\boxed{セソ}}$ である.

(3) $\displaystyle\sum_{n=1}^{\infty} a_n = \dfrac{\boxed{タチ}}{\boxed{ツテ}}$ である.

```
ClearAll[a]
```

```
sol = RSolve[{a[n + 1] == a[n] Cos[n^2 Pi / 3], a[1] == 1}, a[n], n];
a[n_] = (a[n] /. sol)[[1]] // FullSimplify
```

$$(-1)^{1+\text{Floor}\left[\frac{1}{6}(-5+n)\right]+\text{Floor}\left[\frac{1}{6}(-4+n)\right]+\text{Floor}\left[\frac{1}{6}(-3+n)\right]}\ 2^{-3+\text{Ceiling}\left[\frac{2-n}{6}\right]+\text{Ceiling}\left[\frac{3-n}{6}\right]+\text{Ceiling}\left[\frac{5-n}{6}\right]-\text{Floor}\left[\frac{n}{6}\right]}$$

```
(*1*)
a[6]
Sum[a[k], {k, 1, 6}]
```

$-\frac{1}{16}$

$\frac{21}{16}$

```
(*2*)
Log[2, Abs[a[70]]]
Sum[a[6 k + 1], {k, 1, Infinity}]
```

-46

$-\frac{1}{17}$

```
(*3*)
Sum[Sum[a[6 k + r], {k, 0, Infinity}], {r, 1, 6}]
```

$\frac{21}{17}$

4 関数 $f(x)$ を

$$f(x)=\frac{1}{2}x+\frac{1}{8}x^3 \quad (x \geqq 0)$$

により定め，座標平面上の曲線 C を $C: y=f(x)$ とし，曲線 C と y 軸と直線 $y=2$ で囲まれた図形 F について考える．

(1) 図形 F の面積は $\dfrac{\boxed{ア}}{\boxed{イ}}$ である．

(2) 関数 $g(x)$ を $g(x)=f^{-1}(x)^2$ により定義する．このとき $g^{-1}(2)=\dfrac{\boxed{ウ}\sqrt{\boxed{エ}}}{\boxed{オ}}$ である．

ただし，$f^{-1}(x)$ と $g^{-1}(x)$ はそれぞれ $f(x)$ と $g(x)$ の逆関数とする．

(3) 図形 F を y 軸のまわりに回転してできる立体の体積は $\dfrac{\boxed{カキ}}{\boxed{クケ}}\pi$ である．

```
ClearAll[f, x]
f[x_] = 1 / 2 x + 1 / 8 x^3;
F = ImplicitRegion[f[x] < y < 2 && x > 0, {x, y}];
Show[Plot[{f[x], 2}, {x, -1, 3}], RegionPlot[F]]
```



```
(*1*)
RegionMeasure[F]
```

$\frac{5}{2}$

(*2*) (InverseFunction は逆関数の枝のうち適当にひとつを選ぶので,出題者の意図した解答と符号が異なる)

```
ClearAll[f, h, g]
f[x_] = 1 / 2 x + 1 / 8 x^3;
h[x_] = x^2;
g = Composition[h, InverseFunction[f]]; (*g=(f^-1)^2*)
InverseFunction[g][2]
```

$-\frac{3}{2\sqrt{2}}$

(*上では h^{-1}[x]=-$\sqrt{x}$ の枝を取っているので符合が異なる！ そこで手動で解答する*)

```
Composition[f, Sqrt[#] &][2]
```

$\frac{3}{2\sqrt{2}}$

```
(*3*)
(*バウムクーヘン分割の利用*)
Integrate[2 Pi x (2 - f[x]), {x, 0, 2}]
```

$$\frac{56\pi}{15}$$

```
(*RegionMeasure の利用．答えは一致するが，ちょっと時間が掛かる*)
ImplicitRegion[2 > z > f[Sqrt[x^2 + y^2]], {x, y, z}];
RegionMeasure[%]
```

$$\frac{56\pi}{15}$$

5 座標平面上の曲線 $(x^2+y^2)^2 = x^3 - 3xy^2$ を描け．
5 の解答は，数学解答用紙(省略)に解答の曲線だけを記入せよ．解答の曲線以外の補助線や目盛りの数値を新たに記入してはならない．

```
ContourPlot[(x^2 + y^2)^2 == x^3 - 3 x y^2,
 {x, -0.7, 1.1}, {y, -1, 1}, PlotPoints → 50, Axes → True]
```